```
" Force analysis via FBD of individual links "

Apply [Clear, Names["Global`*"] ] ;

Off[General::spell];
Off[General::spell1];

(* Input data *)
pi = N [ Pi ] ;
n = 30 pi; (* rpm *)
omega = n pi/30 ; (* rad/s *)

rule = {AB->0.14, AC->0.06, CF->0.2,
h->0.01, d->0.01,
hSlider->0.02, wSlider->0.05,
ro->8000, g->9.807,
Me->1000.,
phi[t]->pi/3, phi'[t]->omega, phi''[t]->0}


(* Position analysis *)

(* Position of joint A *)
xA = yA = 0 ;
rA = { xA, yA, 0} ;

(* Position of joint C *)
xC = 0 ;
yC = AC ;
rC = { xC, yC, 0} ;

(* Position, velocity and acceleration of joint B *)
xB = AB Cos[phi[t]] ;
yB = AB Sin[phi[t]] ;
rB = { xB, yB, 0} ;
vB = D[rB,t] ;
aB = D[D[rB,t],t] ;

(* Angular velocity and acceleration of the link 1 *)
omega1 = {0, 0, omega} ;
alpha1 = {0, 0, phi''[t]} ;

(* Angular velocity and acceleration of the link 2 and 3 *)

mBC = ( yB - yC ) / ( xB - xC ) ;
phi2 = ArcTan[mBC];
omega2 = {0, 0, D[phi2,t]};
alpha2 = {0, 0, D[D[phi2,t],t]};
phi3=phi2;
alpha3=alpha2;

(*Position, velocity and acceleration of center of mass of link 1 *)
rC1 = rB/2 ;
vC1 = vB/2 ;
aC1 = aB/2 ;

(*Position, velocity and acceleration of center of mass of link 2 *)
rC2 = rB ;
vC2 = vB ;
aC2 = aB ;

(*Position, velocity and acceleration of center of mass of link 3 *)
xC3 = xC + CF/2 Cos[phi3] ;
yC3 = yC + CF/2 Sin[phi3] ;
```

```
rC3 = { xC3 , yC3 , 0 } ;
vC3 = D[rC3,t] ;
aC3 = D[D[rC3,t],t] ;

pxB=xB/.rule;
pyB=yB/.rule;
pxC=xC/.rule;
pyC=yC/.rule;
pxF=(xC+CF Cos[phi3])/.rule;
pyF=(yC+CF Sin[phi3])/.rule;

Print["rB = {xB, yB, 0} = ",rB/.rule," [m]"];
Print["rC1 = {xC1, yC1, 0} = ",rC1/.rule," [m]"];
Print["rC = {xC, yC, 0} = ",rC/.rule," [m]"];
Print["rF = {xF, yF, 0} = ",{pxF,pyF,0}," [m]"];
Print["rC3 = {xC3, yC3, 0} = ",rC3/.rule," [m]"];


markers=
Table[{Point[{0,0}],Point[{pxB,pyB}],Point[{pxC,pyC}],
        Point[{pxF,pyF}]}];

name=
Table[{Text["A",{0,0},{1,1}],Text["B",{pxB,pyB},{-.5,-1.5}],
        Text["C",{pxC,pyC},{1,1}],Text["F",{pxF,pyF},{1,0}]}];

graph[increment]=
    Graphics[{{RGBColor[1,0,0],Line[{{0,0},{pxB,pyB}}]},
    {RGBColor[0,0,0],Line[{{pxC,pyC},{pxF,pyF}}]},
    {RGBColor[0,0,1],PointSize[0.02],
    markers},{name}}];

Show[Graphics[graph[increment]],Frame->True,
    PlotRange->{{-.1,0.3},{-.1,0.3}},AxesOrigin->{0,0},
    FrameLabel->{"x","y"},Axes->{True,True},
    AspectRatio->Automatic];

" Inertia forces and moments "

" Link 1 "
m1 = ro AB h d /.rule ;
IC1 = m1 (AB^2+h^2)/12 /.rule ;
Fin1 = - m1 aC1 /.rule ;
G1 = {0, -m1*g, 0} /.rule ;
F1 = ( Fin1 + G1 ) /.rule ;
M1 = Min1 = - IC1 alpha1 /.rule ;
Print["m1 = ro AB h d = ",m1," [kg]"];
Print["IC1 = m1 (AB^2+h^2)/12 = ",IC1," [kg m^2]"];
Print["aC1 = ",aC1/.rule," [m/s^2]"];
Print["alpha10 = ",alpha1/.rule," [rad/s^2]"];
Print["Fin1 = -m1  aC1 = ",Fin1," [N]"];
Print["F1 = Fin1 + G1 = ",F1," [N]"];
Print["Min1 = M1 = -IC1 alpha10 = ",Min1," [N m]"];

" Link 2 "
m2 = ro hSlider wSlider d /.rule ;
IC2 = m2 (hSlider^2+wSlider^2)/12 /.rule ;
IC2 = 0.;
Fin2 = - m2 aC2 /.rule ;
G2 = {0, -m2*g, 0} /.rule ;
F2 = ( Fin2 + G2 ) /.rule ;
M2 = Min2 = - IC2 alpha2 /.rule ;
Print["m2 = ro hSlider wSlider d = ",m2," [kg]"];
Print["IC2 = m2 (hSlider^2+wSlider^2)/12 = ",IC2," [kg m^2]"];
Print["aC2 = ",aC2 /.rule," [m/s^2]"];
```

```
Print["alpha20 = ",alpha2/.rule," [rad/s^2]"];
Print["Fin2 = -m2  aC2 = ",Fin2," [N]"];
Print["F2 = Fin2 + G2 = ",F2," [N]"];
Print["Min2 = M2 = -IC2 alpha20 = ",Min2," [N m]"];

" Link 3 "
m3 = ro CF h d /.rule ;
IC3 = m3 (CF^2+h^2)/12 /.rule ;
Fin3 = - m3 aC3 /.rule ;
G3 = {0, -m3*g, 0} /.rule ;
F3 = ( Fin3 + G3 ) /.rule ;
M3 = Min3 = - IC3 alpha3 /.rule ;
Print["m3 = ro CF h d = ",m3," [kg]"];
Print["IC3 = m3 (CF^2+h^2)/12 = ",IC3," [kg m^2]"];
Print["aC3 = ",aC3 /.rule," [m/s^2]"];
Print["alpha30 = ",alpha3 /.rule," [rad/s^2]"];
Print["Fin3 = -m3  aC3 = ",Fin3," [N]"];
Print["F3 = Fin3 + G3 = ",F3," [N]"];
Print["Min3 = M3 = -IC3  alpha30 =",Min3," [N m]"];

M3e = -Sign[omega2]{ 0, 0, Me } /.rule ;
Print["M3e = -Sign[omega2]{ 0, 0, Me } = ",M3e," [N m]"];

" Joint reactions "

" Link 3 "
F03 = { F03x, F03y, 0 } ;
F23 = { F23x, F23y, 0 } ;
rQ = {xQ , yQ,0};

Print["Sum F for link 3: F3 + F03 + F23 = 0"];
eqF3=(F3+F03+F23);
Print["(x): ",eqF3[[1]],"=0 (1)"];
Print["(y): ",eqF3[[2]],"=0 (2)"];
Print["Sum M for link 3 wrt C3:  rC3C x F03 + C3Q x F23 + M3 + M3e = 0"];
rC3C = ( rC - rC3 ) /.rule ;
rC3Q = ( rQ - rC3 ) /.rule ;
eqM3=Cross[rC3C,F03]+Cross[rC3Q,F23]+M3+M3e;
Print["(z): ",eqM3[[3]],"=0 (3)"];

Print["F23 perpendicular to BC: F32.BC = 0 => "]
rBC = ( rC - rB )/.rule ;
eqBT1=F23.rBC==0;
Print[F32.rBC,"=0 (4)"];
Print["point Q is on BC: (yB-yC)/(xB-xC)=(yQ-yC)/(xQ-xC) => "]
eqBT2=((yB-yC)/(xB-xC)/.rule)==((yQ-yC)/(xQ-xC)/.rule);
Print[((yB-yC)/(xB-xC)/.rule)-((yQ-yC)/(xQ-xC)/.rule),"=0 (5)"];

" Link 2 "
F32 =-F23;
F12 = { F12x, F12y, 0 } ;
Print["Sum F for link 2: F2 + F32 + F12 = 0"];
eqF2=(F2+F32+F12) ;
Print["(x): ",eqF2[[1]],"=0 (6)"];
Print["(y): ",eqF2[[2]],"=0 (7)"];
Print["Sum M for link 2 wrt C2(B): BQ x F32 + M2 = 0"];
rBQ = ( rQ - rB ) /.rule ;
eqM2=Cross[rBQ,F32]+M2;
Print["(z): ",eqM2[[3]],"=0 (8)"];

solF=Solve[{eqF3[[1]]==0,eqF3[[2]]==0,eqM3[[3]]==0,eqBT1,
        eqBT2,eqF2[[1]]==0,eqF2[[2]]==0,eqM2[[3]]==0},
        {F03x,F03y,F23x,F23y,F12x,F12y,xQ,yQ}];

Print["From Eqs.(1)-(8) => F03x, F03y, F23x, F23y, F12x, F12y, xQ, yQ"]
```

```
Print["F03 = {F03x, F03y, 0} = ",F03/.solF[[1]]," [N]"];
Print["F23 = {F23x, F23y, 0} = ",F23/.solF[[1]]," [N]"];
Print["F12 = {F12x, F12y, 0} = ",F12/.solF[[1]]," [N]"];
Print["rQ = {xQ, yQ, 0} = ",rQ/.solF[[1]]," [m]"];

 " Link 1 "
F21 =-F12/.solF[[1]];
F01={F01x,F01y,0};
M1m={0,0,Mm};
Print["Sum F for link 1: F1 + F21 + F01 = 0 <=> F01 = -F21 - F1"];
F01S=-(F1+F21);
Print["F01 = ",-F1-F21," [N]"];
Print["Sum M for 1 wrt C1:  C1B x F21 + C1A x F01 + M1 + Mm = 0 <=>"];
Print["Mm = - ( C1B x F21 + C1A x F01 + M1 ) "];
rC1B=(rB-rC1)/.rule;
rC1A=(-rC1)/.rule;
M1m=-Cross[rC1B,F21]-Cross[rC1A,F01S]-M1;
Print["Mm = ",M1m," [Nm]"];
```

Force analysis via FBD of individual links

```
{AB → 0.14, AC → 0.06, CF → 0.2, h → 0.01, d → 0.01, hSlider → 0.02, wSlider → 0.05,
 ro → 8000, g → 9.807, Me → 1000., phi[t] → 1.0472, phi′[t] → 9.8696, phi″[t] → 0}

rB = {xB, yB, 0} = {0.07, 0.121244, 0} [m]

rC1 = {xC1, yC1, 0} = {0.035, 0.0606218, 0} [m]

rC = {xC, yC, 0} = {0, 0.06, 0} [m]

rF = {xF, yF, 0} = {0.150522, 0.191693, 0} [m]

rC3 = {xC3, yC3, 0} = {0.0752611, 0.125847, 0} [m]
```



Inertia forces and moments

Link 1

```
m1 = ro AB h d = 0.112 [kg]
```

IC1 = m1 (AB^2+h^2)/12 = 0.000183867 [kg m^2]

aC1 = {-3.40932, -5.90511, 0} [m/s^2]

alpha10 = {0, 0, 0} [rad/s^2]

Fin1 = -m1 aC1 = {0.381844, 0.661373, 0} [N]

F1 = Fin1 + G1 = {0.381844, -0.437011, 0} [N]

Min1 = M1 = -IC1 alpha10 = {0, 0, 0} [N m]

Link 2

m2 = ro hSlider wSlider d = 0.08 [kg]

IC2 = m2 (hSlider^2+wSlider^2)/12 = 0. [kg m^2]

aC2 = {-6.81864, -11.8102, 0} [m/s^2]

alpha20 = {0, 0, 87.47} [rad/s^2]

Fin2 = -m2 aC2 = {0.545491, 0.944818, 0} [N]

F2 = Fin2 + G2 = {0.545491, 0.160258, 0} [N]

Min2 = M2 = -IC2 alpha20 = {0, 0, 0.} [N m]

Link 3

m3 = ro CF h d = 0.16 [kg]

IC3 = m3 (CF^2+h^2)/12 = 0.000534667 [kg m^2]

aC3 = {-20.6416, -6.4373, 0} [m/s^2]

alpha30 = {0, 0, 87.47} [rad/s^2]

Fin3 = -m3 aC3 = {3.30266, 1.02997, 0} [N]

F3 = Fin3 + G3 = {3.30266, -0.539153, 0} [N]

Min3 = M3 = -IC3 alpha30 ={0, 0, -0.0467673} [N m]

M3e = -Sign[omega2]{ 0, 0, Me } = {0, 0, -1000.} [N m]

Joint reactions

Link 3

Sum F for link 3: F3 + F03 + F23 = 0

(x): $3.30266 + F03x + F23x = 0$ (1)

(y): $-0.539153 + F03y + F23y = 0$ (2)

Sum M for link 3 wrt C3: rC3C x F03 + C3Q x F23 + M3 + M3e = 0

(z): $-1000.05 + 0.0658465\, F03x - 0.0752611\, F03y + 0.125847\, F23x - 0.0752611\, F23y + F23y\, xQ - F23x\, yQ = 0$ (3)

F23 perpendicular to BC: F32.BC = 0 =>

F32.{-0.07, -0.0612436, 0}=0 (4)

point Q is on BC: (yB-yC)/(xB-xC)=(yQ-yC)/(xQ-xC) =>

$$0.874908 - \frac{-0.06 + yQ}{xQ} = 0 \quad (5)$$

Link 2

Sum F for link 2: F2 + F32 + F12 = 0

(x): 0.545491 + F12x - F23x=0 (6)

(y): 0.160258 + F12y - F23y=0 (7)

Sum M for link 2 wrt C2(B): BQ x F32 + M2 = 0

(z): 0. - 0.121244 F23x + 0.07 F23y - F23y xQ + F23x yQ=0 (8)

From Eqs.(1)-(8) => F03x, F03y, F23x, F23y, F12x, F12y, xQ, yQ

F03 = {F03x, F03y, 0} = {7078.4, -8093.69, 0} [N]

F23 = {F23x, F23y, 0} = {-7081.71, 8094.23, 0} [N]

F12 = {F12x, F12y, 0} = {-7082.25, 8094.07, 0} [N]

rQ = {xQ, yQ, 0} = {0.07, 0.121244, 0} [m]

Link 1

Sum F for link 1: F1 + F21 + F01 = 0 <=> F01 = -F21 - F1

F01 = {-7082.63, 8094.51, 0} [N]

Sum M for 1 wrt C1: C1B x F21 + C1A x F01 + M1 + Mm = 0 <=>

Mm = - (C1B x F21 + C1A x F01 + M1)

Mm = {0., 0., 1425.3} [Nm]